
****************************************************************

* Echo Imitator                                                                                                     *

* Input: Transmitter pulse                                                                                     *

* Output:Pulse 1 imitates 2ms ringing of the receiver sensor   and                      *

*              Pulse 2 imitates a  0.5ms echo signal at a variable distance                 *

****************************************************************

.nolist

$include "fr908QT2.asm"    ; Frame file of uC 908QT

.list

.pagelength 80T

.pagewidth 100T

*********** I/O PORTS  *********************

inp    equ  pA0 ;input  transmitter pulse

out    equ  pA1 ;output echo pulse

pot    equ  pA4 ;input  ADC2 set Distance

********* VARIABLES   ***********************

          org     RAM

DSTcnt    rmb     1 ;Distance counter

********** INITIALIZATION    ****************

           org     ROM

init    rsp                 ;reset stack pointer to $ff

         lda     $ffc0    ;adjust frequency to 12.5MHz+/-5%

         sta      $0038

         mov    #$01,CONFIG1    ;COP disabled

         clrA

         clrX

         clrH

         clr       DSTcnt

* Set I/O

        clr        prtA

        mov     #%00000010,ddrA ; pA0,pA4 –inputs,  pA1-output,

* Set Timer 

        mov     #%01110100,TSC ;TOFInt enable,Stop and reset Timer,

        clr       TmodH                   ;                       prescaler=16

        mov     #100T,TmodL       ;set TOF every 0.5 ms

        bclr      TSTOP,TSC          ;start Timer counter

        cli                                       ;set interrupts enable

.page

**************  MAIN PROGRAM **********************************

* Wait for the transmitter pulse front edge

* When it comes, generate 2ms "bouncing sensor" pulse

* Read the Distance set by potentiometer

* Wait until incremented DST counter becomes equal to DST set

* Generate 0.5ms "echo" pulse

* Return to Main

**************************************************************

Main    brclr   inp,prtA,*      ;wait for input tr =1

             clr      DSTcnt          ;0-->DSTcnt

*******  Generate 2 ms pulse  *******************

             bset    out,prtA        ;start pulse

             lda     DSTcnt          ;DSTcnt-->Acc

             add     #4T               ;Acc+4-->Acc

m0        cmp    DSTcnt         ;repeat until Acc=DSTcnt,

             bne     m0

             bclr    out,prtA         ;end pulse

******* Read ADC2 for DST setting  **************

             mov    #%00000010,ADscr  ;start ADC2 on pA4

             brclr   COCO,ADscr,*          ;wait for conversion end

             mov    #%00011111,ADscr  ;stop ADC

******* Find when DST setting = DSTcnt  *********

              lda     ADR              ;ADC Data-->Acc

              lsrA                          ;divide by 4

              lsrA

              cmp     #6T
    ;if Acc<6, exit

              blo      m3

m1         cmp     DSTcnt         ;repeat till   Acc=DSTcnt

              bhi       m1

******* Generate  0.5ms echo pulse  *************

              bset      out,prtA

              lda       DSTcnt

              add       #1

m2         cmp     DSTcnt

              bhi       m2

              bclr      out,prtA

m3         bra       Main

************* TOF Interrupt Service Routine   *******************

TimInt    inc      DSTcnt

               lda      TSC        ;TOF reset

               bclr     TOF,TSC

               rti

*****************************************************************

        org   $fff2

        fdb   TimInt

        org   RESET

        fdb   init

.end

.nolist

